	[bookmark: _Toc306786316][bookmark: _Toc373917918]Web Appendix 1:  Food coding schemea for aggregating foods from National Health and Nutrition Examination Survey, 2007-2010, into fruit and vegetable groups comparable to Behavioral Risk Factor Surveillance System (BRFSS) 2011

	2011 BRFSS Question
	Food Codes Includedb

	1. During the past month, how many times per day, week or month did you drink 100% PURE fruit juices? Do not include fruit-flavored drinks with added sugar or fruit juice you made at home and added sugar to. Only include 100% juice. 

	611-----   or 61201--- thru 61213--- or 61216---
and  6120050- or 61214--- or 61219--- thru 61226---- or 6410011-  thru  6422101-

	2. During the past month, not counting juice, how many times per day, week, or month did you eat fruit? Count fresh, frozen, or canned fruit.

	6110---- thru 634-----  or 641----- thru 642----- 
excluding fruit juice (see above)

	3. During the past month, how many times per day, week, or month did you eat cooked or canned beans, such as refried, baked, black, garbanzo beans, beans in soup, soybeans, edamame, tofu or lentils. Do NOT include long green beans. 

	41------or 1131---- thru 1133---- or 1172----

	4. During the past month, how many times per day, week, or month did you eat dark green vegetables for example broccoli or dark leafy greens including romaine, chard, collard greens or spinach?

	72------ or 75147---


	5. During the past month, how many times per day, week, or month did you eat orange- colored vegetables such as sweet potatoes, pumpkin, winter squash, or carrots?

	73------


	6. Not counting what you just told me about, during the past month, about how many times per day, week, or month did you eat OTHER vegetables? Examples of other vegetables include tomatoes, tomato juice or V-8 juice, corn, eggplant, peas, lettuce, cabbage, and white potatoes that are not fried such as baked or mashed potatoes. 
	7------- excluding dark green vegetables and orange vegetables (see above),  and fried potatoes (712-----or 714-----or 71505--- or 77121---)

	a Based on United State Department of Agriculture, Food Survey Research Group defined food groups.(1,2) To facilitate comparisons with BRFSS certain food codes were as excluded: baby foods (67- ----- and 76- -----), dried fruit (62- -----), fruits and vegetables eaten in combination with sandwiches, and condiments including tomato sauces (salsa, ketchup, spaghetti sauce, etc., 744 -----), olives, pickles, and relishes (755 -----), and vinegar (644 -----).  

 bUnique 8-digit USDA food code assigned to each food and beverage reported by National Health and Nutrition Examination survey participants during 24 dietary recalls.(1,2)




	[bookmark: _Toc373917915]Web Table 1: Regression Coefficients for Models Predicting the Log Odds that Cup Equivalents of Fruits And Vegetables Consumed  ≥ United States Department of Agriculture Food Patterns Intake Recommendations, by Age Group and Sex, National Health and Nutrition Examination Survey, United States, 2007-2010a 

	
	
	Females
	Males

	Parameter
	Description
	18-30
	31-50
	51-70
	71+
	18-30
	31-50
	51-70
	71+

	Fruits
	
	
	
	
	
	
	
	
	

	β0
	Intercept
	-6.0600
	-4.6596
	-4.0696
	-5.4087
	-5.6210
	-5.4948
	-5.4824
	-5.2136

	
	Times per day
	
	
	
	
	
	
	
	

	β1
	Fruit juice
	1.8075
	1.4798
	1.6626
	2.2270
	2.2377
	2.1261
	2.0233
	1.5826

	β2
	Fruit
	1.4916
	1.6647
	1.5286
	1.6380
	1.5963
	1.4769
	1.4799
	1.5583

	
	Race/ethnicity 
	
	
	
	
	
	
	
	

	β3
	Hispanic
	0.3152
	0.7545
	-0.1050
	0.3825
	0.5780
	0.5608
	0.0313
	-0.3240

	β4
	Non-Hispanic Black
	0.3618
	0.2488
	-0.3630
	-0.4413
	0.2255
	0.2916
	-0.1186
	-0.3004

	
	Poverty-income Ratio
	
	
	
	
	
	
	
	

	β5
	<1.25
	-0.4813
	-0.5005
	-0.5196
	0.0331
	-0.4155
	0.0125
	-0.3753
	-0.4122

	β6
	1.25–3.49
	-0.3320
	-0.3023
	-0.6026
	0.3026
	-0.2661
	-0.0540
	-0.6194
	-0.6227

	Vegetables
	
	
	
	
	
	
	
	
	

	β0
	Intercept
	-10.6741
	-7.2551
	-5.0954
	-7.3053
	-8.8961
	-7.2230
	-5.7497
	-7.8390

	
	Times per day
	
	
	
	
	
	
	
	

	β1
	Beans
	3.0741
	1.7454
	1.6233
	1.8603
	2.6907
	2.3002
	1.9619
	2.7171

	β2
	Dark green vegetables
	2.9105
	1.5974
	1.4143
	1.8111
	0.8994
	1.7541
	1.4864
	2.3613

	β3
	Orange vegetables
	1.8676
	1.4572
	0.9277
	1.2978
	1.8565
	1.1451
	0.8856
	0.9653

	β4
	Other vegetables
	2.4543
	1.7268
	1.5122
	1.8868
	2.3095
	1.9565
	1.5990
	1.7729

	
	Race/ethnicity
	
	
	
	
	
	
	
	

	β5
	Hispanic
	1.0553
	0.3476
	-0.2125
	-0.5384
	0.6188
	-0.0590
	-0.0260
	-0.8822

	β6
	Non-Hispanic Black
	-0.5016
	-0.6758
	-0.6900
	-0.7520
	-1.1320
	-1.3282
	-0.5869
	-1.3303

	
	Poverty-income Ratio
	
	
	
	
	
	
	
	

	β7
	<1.25
	-0.9968
	-0.7650
	-0.7254
	-0.7311
	-0.4317
	-0.4398
	-0.8123
	-0.5660

	β8
	1.25–3.49
	-0.3261
	-0.6004
	-0.4627
	-0.4749
	0.0101
	-0.4741
	-0.5782
	-0.1485

	a Models account for true usual intakes of fruits and vegetables,  sequence effect, weekend effect, race/ethnicity, and poverty-income ratio. Intake recommendations are sex and age specific and assume physical activity is < 30 minutes daily.




	Web Table 2: Regression Coefficients for Models Predicting the Cup Equivalents of Fruits And Vegetables Consumed, by Age Group and Sex, National Health and Nutrition Examination Survey, United States, 2007-2010a 

	
	
	Females
	Males

	Parameter
	Description
	18-30
	31-50
	51-70
	71+
	18-30
	31-50
	51-70
	71+

	Fruits
	
	
	
	
	
	
	
	
	

	β0
	Intercept
	0.2063
	0.1997
	0.3655
	0.1604
	0.1972
	0.1864
	0.3115
	0.3659

	
	Times per day
	
	
	
	
	
	
	
	

	β1
	Fruit juice
	0.5396
	0.4840
	0.4711
	0.5535
	0.7503
	0.7343
	0.6556
	0.5312

	β2
	Fruit
	0.4298
	0.5688
	0.5372
	0.5089
	0.5699
	0.6267
	0.5285
	0.5683

	
	Race/ethnicity 
	
	
	
	
	
	
	
	

	β3
	Hispanic
	0.0454
	0.1285
	-0.0257
	0.0935
	0.1495
	0.1404
	0.0092
	-0.1160

	β4
	Non-Hispanic Black
	0.0646
	0.0465
	-0.1052
	-0.0556
	0.0299
	0.0610
	0.0164
	-0.0606

	
	Poverty-income Ratio
	
	
	
	
	
	
	
	

	β5
	<1.25
	-0.0293
	-0.0687
	-0.1092
	0.0313
	-0.0152
	-0.0079
	-0.1048
	-0.0769

	β6
	1.25–3.49
	-0.0100
	-0.0434
	-0.0990
	0.0832
	-0.0341
	-0.0282
	-0.1219
	-0.1613

	Vegetables
	
	
	
	
	
	
	
	
	

	β0
	Intercept
	0.5293
	0.6299
	0.7186
	0.6722
	0.8281
	0.7765
	0.8401
	0.6269

	
	Times per day
	
	
	
	
	
	
	
	

	β1
	Beans
	0.5272
	0.4379
	0.4175
	0.3676
	0.6459
	0.8767
	0.6016
	0.6651

	β2
	Dark green vegetables
	0.6668
	0.5614
	0.4234
	0.3864
	0.2058
	0.6341
	0.4451
	0.5906

	β3
	Orange vegetables
	0.3365
	0.4207
	0.2673
	0.2038
	0.4556
	0.3672
	0.2616
	0.2249

	β4
	Other vegetables
	0.4884
	0.4972
	0.4304
	0.3760
	0.6147
	0.6960
	0.5264
	0.4595

	
	Race/ethnicity
	
	
	
	
	
	
	
	

	β5
	Hispanic
	0.1346
	0.0257
	-0.0273
	-0.0820
	0.1209
	-0.0962
	-0.0302
	-0.1292

	β6
	Non-Hispanic Black
	-0.0268
	-0.1288
	-0.1449
	-0.1330
	-0.1694
	-0.3187
	-0.1647
	-0.2519

	
	Poverty-income Ratio
	
	
	
	
	
	
	
	

	β7
	<1.25
	-0.0727
	-0.1140
	-0.1728
	-0.1368
	-0.1301
	-0.0449
	-0.1557
	-0.1033

	β8
	1.25–3.49
	-0.0170
	-0.1082
	-0.1041
	-0.0906
	-0.0445
	-0.0741
	-0.1136
	-0.0054

	a Models account for true usual intakes of fruits and vegetables,  sequence effect, weekend effect, race/ethnicity, and poverty-income ratio. 











Web Appendix 2: SAS Callable SUDAAN code to estimate median number of times fruits and vegetables reported per day and percent of the population meeting USDA Food Patterns fruit and vegetable intake recommendations, Behavioral Risk Factor Surveillance System (BRFSS), 2011

*********************************************************************************
Program: 	
Web Appendix 2 SAS code
												
Written By:	
LV Moore, PhD MSPH | Epidemiologist 							
Division of Nutrition, Physical Activity, and Obesity				
National Center for Chronic Disease Prevention and Health Promotion 	
Centers for Disease Control & Prevention						
Email: lvmoore@cdc.gov 											
																			
 Date:
03/2014 
														
Input Dataset:
LLCP2011.XPT
													
Output Dataset: 
Work.estimates
													
 Description:																	
 This program generates national and state specific estimates of fruit and 	
 vegetable intake from 2011 BRFSS. Six indicators are created: 				
																			
	1) median times per day fruit are consumed								
	2) median times per day vegetables are consumed 						
	3) percent of the fruit recommendation the population met				
	4) percent of the vegetable recommendation the population met			
	5) percent of the population meeting fruit recommendations				
	6) percent of the population meeting vegetable recommendations	
*********************************************************************************;

*********************************************************************************
NOTE: Set location of 2011 BRFSS XPORT file 
*********************************************************************************;
	%let library=\\xxx\xxx\xxx;
	LIBNAME BRFSS11 XPORT "&library\LLCP2011.XPT";

*********************************************************************************
SAS code to estimate median intake of fruits and vegetables (times per day)
*********************************************************************************;
	DATA one;
		SET brfss11.LLCP2011 (KEEP=_PSU _STSTR _LLCPWT SEX _IMPAGE _imprace  
							NUMADULT CHILDREN INCOME2 FRUITJU1 FRUIT1 FVBEANS 
							FVGREEN FVORANG VEGETAB1 _state _FRUITEX _VEGETEX);
	
	*Algorithm to estimate times per day fv consumed;
		ARRAY FVIN  [6] FRUITJU1 FRUIT1 FVBEANS FVGREEN FVORANG	VEGETAB1; 
		ARRAY FVOUT [6] FTJUDA1_ FRUTDA1_ BEANDAY_ GRENDAY_ ORNGDAY_ VEGEDA1_ ;
		DO I=1 TO 6;
			IF 101<=FVIN(I)<=199 	  THEN FVOUT(I)=FVIN(I)-100;
		    	ELSE IF 201<=FVIN(I)<=299 THEN FVOUT(I)=(FVIN(I)-200)/7;
				ELSE IF 301<=FVIN(I)<=399 THEN FVOUT(I)=(FVIN(I)-300)/30; 
				ELSE IF FVIN(I)=300	     THEN FVOUT(I)=.02; 
		        ELSE IF 401<=FVIN(I)<=499 THEN FVOUT(I)=(FVIN(I)-400)/365; 
				ELSE IF FVIN(I)=555 	  THEN FVOUT(I)=0; 
		     	ELSE IF FVIN(I)in (777,999,.) THEN FVOUT(I)=.;
		END;

		*Calculate total fruit and total vegetable intake;
		_FRUTSUM=FTJUDA1_+FRUTDA1_;
		_VEGESUM=BEANDAY_ + GRENDAY_ + ORNGDAY_ + VEGEDA1_;

	*Create demographic variables needed for algorithms to estimate cup 
	equivalents and percent meeting recommendations;

		*Variable 1:sexagec - 8 groups in the ChooseMyPlate.gov Fruit and Vegetable 
			Daily Recommendations;
			select;
				*Females;
					 when (_IMPAGE >= 18 and _IMPAGE <= 30 and SEX=2) sexagec=1;
					 when (_IMPAGE >= 31 and _IMPAGE <= 50 and SEX=2) sexagec=2;
					 when (_IMPAGE >= 51 and _IMPAGE <= 70 and SEX=2) sexagec=3;
					 when (_IMPAGE >= 71 and SEX=2) sexagec=4;
				*Males;
				    when (_IMPAGE >= 18 and _IMPAGE <= 30 and SEX=1) sexagec=5;
				    when (_IMPAGE >= 31 and _IMPAGE <= 50 and SEX=1) sexagec=6;
				    when (_IMPAGE >= 51 and _IMPAGE <= 70 and SEX=1) sexagec=7;
				    when (_IMPAGE >= 71 and SEX=1) sexagec=8;
				otherwise sexagec=.;
			end;

		*Variable 2-3: Binary race/ethnicity variables;
			if _imprace ne . then do;
				race1=(_imprace=5);*Hispanic;
				race2=(_imprace=2);*NH Black;
			end;

		*Variable 4-6: Binary poverty to income ratio variables using 
			ASPE 2011 poverty guidelines;
			*Calculating TOTAL HOUSEHOLD SIZE FROM NUMBER OF ADULTS & NUMBER OF CHILDREN;
				*Number of children capped at 18;
				IF CHILDREN IN (., 88,99) THEN CHILDREN_CAPPED=0;
					ELSE IF  CHILDREN>18 THEN CHILDREN_CAPPED=18;
					ELSE IF 1<=CHILDREN<=18 THEN CHILDREN_CAPPED=CHILDREN;
				*IF RESPONDENT DID NOT REPORT NO. OF ADULTS ASSUME 1 ADULT;
				IF NUMADULT=. THEN NUMADULTR=1;
					ELSE NUMADULTR=NUMADULT;
				THH_SIZE=NUMADULTR+CHILDREN_CAPPED;
			*HOUSEHOLD SIZE capped at 7 (99TH PERCENTILE OF HOUSEHOLD IN NHANES);
				IF 1<=THH_SIZE<=6 THEN THH_PIR=THH_SIZE;
					ELSE IF THH_SIZE > 6 THEN THH_PIR=7;
			*Income level by household size from ASPE Tables; 
				IF THH_PIR=1 THEN INCLEVEL2=10890;
					ELSE IF THH_PIR=2 THEN INCLEVEL2=14710;
					ELSE IF THH_PIR=3 THEN INCLEVEL2=18530;
					ELSE IF THH_PIR=4 THEN INCLEVEL2=22350;
					ELSE IF THH_PIR=5 THEN INCLEVEL2=26170;
					ELSE IF THH_PIR=6 THEN INCLEVEL2=29990;
					ELSE IF THH_PIR=7 THEN INCLEVEL2=33810;
					ELSE INCLEVEL2=.;
			*Use MIDPOINT OF INCOME2 RANGE TO CREATE PIR;
				IF INCOME2=1 THEN INDFMPIR=5000/INCLEVEL2; /*LESS THAN 10000*/
					ELSE IF INCOME2=2 THEN INDFMPIR=12500/INCLEVEL2; /*10000 - <15000*/
					ELSE IF INCOME2=3 THEN INDFMPIR=17500/INCLEVEL2; /*15000 - <20000*/
					ELSE IF INCOME2=4 THEN INDFMPIR=22500/INCLEVEL2; /*20000 - <25000*/
					ELSE IF INCOME2=5 THEN INDFMPIR=30000/INCLEVEL2; /*25000 - <35000*/
					ELSE IF INCOME2=6 THEN INDFMPIR=42500/INCLEVEL2; /*35000 - <50000*/
					ELSE IF INCOME2=7 THEN INDFMPIR=62500/INCLEVEL2; /*50000 - <75000*/
					ELSE IF INCOME2=8 THEN INDFMPIR=87500/INCLEVEL2; /*>=75000*/
					ELSE INDFMPIR=.;
				if INDFMPIR=. THEN PIR=.;
					ELSE IF 0 LE INDFMPIR lt 1.25 then PIR=1;*<125% poverty;
					ELSE if 1.25 le INDFMPIR le 3.49 then PIR=2;*125%-349%;
					else IF INDFMPIR GT 3.49 THEN PIR=3;*PIR>349%
			*PIR binary variables for analysis;
				if PIR ne . then do;
					PIR1=(PIR=1 AND PIR NE .);
					PIR2=(PIR=2 AND PIR NE .);
				end;

	*Assign recommended intake based on age and sex from the ChooseMyPlate.gov
		Fruit and Vegetable Daily Recommendations;
		*Fruit;
		select;
		    when (_impage = 18 and sex=2) f_rec=1.5;
		    when (19 le _impage le 30 and sex=2) f_rec=2 ;
		    when (_impage ge 31 and sex=2) f_rec=1.5 ;
		    when (sex=1) f_rec=2;
		end;
		*Vegetables;
		select;
		    when (18 le _impage le 50 and sex=2) v_rec=2.5;
		    when (_impage ge 51 and sex=2) v_rec=2;
		    when (18 le _impage le 50 and sex=1) v_rec=3;
		    when (_impage ge 51 and sex=1) v_rec=2.5;
		end;
	*Flag records with missing fruit, vegetable, or demographic information and implausible 
		fv values (f>16 times per day or v > 23 times per day);
		complete=(sexagec ne . and _FRUITEX=0 and _VEGETEX=0 and pir1 ne . 
			and pir2 ne . and race1 ne . and race2 ne . and _state le 56);

	label FTJUDA1_='Fruit juice intake in times per day'
		FRUTDA1_='Fruit intake in times per day'
		BEANDAY_='Bean intake in times per day'
		GRENDAY_='Dark green vegetable intake in times per day'
		ORNGDAY_='Orange-colored vegetable intake in times per day'
		VEGEDA1_='Other vegetable intake in times per day'
		_FRUTSUM='Total fruits consumed per day'
		_VEGESUM='Total vegetables consumed per day'
		sexagec='Participant sex age group in the ChooseMyPlate.gov Recommendations'
		race1='Binary race/ethnicity variable: 1=Hisp'
		race2='Binary race/ethnicity variables: 1=NH Blk'
		pir1='Binary poverty to income ratio variable: 1=PIR < 1.25'
		pir2='Binary poverty to income ratio variable: 1=1.25 <= PIR <=3.49'
		f_rec='Sex and age specific recommended fruit intake'
		v_rec='Sex and age specific recommended vegetable intake'
		complete='Participants w/ complete & plausible fv & demographic data:1=No missing'; 
run;
	*SAS Callable SUDAAN code to generate national and state median times per day;
	proc sort data=one; by _STSTR _PSU;
	proc descript data=one filetype=sas design=wr;
		NEST  _STSTR _PSU/missunit;
		weight _LLCPWT;
		subpopn complete=1/name="Respondents with no missing fv or demographic information";
		var _FRUTSUM _VEGESUM ;
		tables 	_state ;
		class _state  ;
		Percentile / median ;
		output nsum  qtile=median /filetype=sas filename=work.temp  replace;
	run;
	data times_per_day;
		set temp;
		length fv $20.;
		if variable=2 then fv='TimesVegetable'; else fv='TimesFruit';
		if _state ne 0 then statename=fipname(_state); else statename='National';
		keep _state statename fv nsum median;
	run;
*********************************************************************************
 Estimate percent of the recommendations met			      
*********************************************************************************;

	*Use conversion equation regression coefficients to estimate amount of fruits
		and vegetables each respondent consumed;
	*Regression parameters for amount of fruit consumed models;
	data parmsrecf_amount;
		input sexagec model $9. Intercept f_juice_intake f_wholefrt_intake race1 race2 pir1 pir2;
		cards;
	1 E(yvar_f) 0.2063 0.5396	0.4298	0.0454	0.0646	-0.0293	-0.01
	2 E(yvar_f) 0.1997 0.484 0.5688 0.1285 0.0465 -0.0687 -0.0434
	3 E(yvar_f) 0.3655 0.4711 0.5372 -0.0257 -0.1052 -0.1092 -0.099
	4 E(yvar_f) 0.1604 0.5535 0.5089 0.0935 -0.0556 0.0313 0.0832
	5 E(yvar_f) 0.1972 0.7503 0.5699 0.1495 0.0299 -0.0152 -0.0341
	6 E(yvar_f) 0.1864 0.7343 0.6267 0.1404 0.061 -0.0079 -0.0282
	7 E(yvar_f) 0.3115 0.6556 0.5285 0.0092 0.0164 -0.1048 -0.1219
	8 E(yvar_f) 0.3659 0.5312 0.5683 -0.116 -0.0606 -0.0769 -0.1613
;
 *Regression parameters for amount of vegetables consumed models;
	data parmsrecv_amount;
		input sexagec model $9. Intercept v_leg_intake v_drkgr_intake v_orange_intake v_other_intake race1
		race2 pir1 pir2;
		cards;
	1 E(yvar_v) 0.5293 0.5272 0.6668 0.3365 0.4884 0.1346 -0.0268 -0.0727 -0.017
	2 E(yvar_v) 0.6299 0.4379 0.5614 0.4207 0.4972 0.0257 -0.1288 -0.114 -0.1082
	3 E(yvar_v) 0.7186 0.4175 0.4234 0.2673 0.4304 -0.0273 -0.1449 -0.1728 -0.1041
	4 E(yvar_v) 0.6722 0.3676 0.3864 0.2038 0.376 -0.082 -0.133 -0.1368 -0.0906
	5 E(yvar_v) 0.8281 0.6459 0.2058 0.4556 0.6147 0.1209 -0.1694 -0.1301 -0.0445
	6 E(yvar_v) 0.7765 0.8767 0.6341 0.3672 0.696 -0.0962 -0.3187 -0.0449 -0.0741
	7 E(yvar_v) 0.8401 0.6016 0.4451 0.2616 0.5264 -0.0302 -0.1647 -0.1557 -0.1136
	8 E(yvar_v) 0.6269 0.6651 0.5906 0.2249 0.4595 -0.1292 -0.2519 -0.1033 -0.0054
;
run;
	proc sql;     
		create table two as
		select a._state, a.sexagec, a.FTJUDA1_, a.FRUTDA1_, a.BEANDAY_,
			a.GRENDAY_, a.ORNGDAY_, a.VEGEDA1_,_PSU, _STSTR, _LLCPWT, a.complete,
				b.intercept + (b.f_juice_intake*FTJUDA1_) 
				+ (b.f_wholefrt_intake*FRUTDA1_) + (b.race1*a.race1)
				+ (b.race2*a.race2)+ (b.pir1*a.pir1) + (b.pir2*a.pir2)
				as amountf label='Estimated amount of fruit consumed',

			c.intercept + (c.v_leg_intake*BEANDAY_) 
				+ (c.v_drkgr_intake*GRENDAY_) +(c.v_orange_intake*ORNGDAY_) 
				+ (c.v_other_intake*VEGEDA1_)+ (c.race1*a.race1)  
				+ (c.race2*a.race2)+ (c.pir1*a.pir1) + (c.pir2*a.pir2)
				as amountv label='Estimated amount of vegetables consumed',

			calculated amountf/f_rec as percentmetf label='Percent of fruit recommendation met', 
			calculated amountv/v_rec as percentmetv label='Percent of vegetable recommendation met'

		from one as a,  parmsrecf_amount as b, parmsrecv_amount as c
		where a.sexagec=b.sexagec=c.sexagec
		order by a.sexagec;
	quit;

	*SAS Callable SUDAAN code to estimate national and state percent meeting recommendations;
	proc sort data=two; by _STSTR _PSU;
	proc descript data=two filetype=sas design=wr;
		NEST  _STSTR _PSU/missunit;
		weight _LLCPWT;
		subpopn complete=1/name="Respondents with no missing fv or demographic information";
		var percentmetf percentmetv ;
		tables 	_state ;
		class _state  ;
		output nsum mean /filetype=sas filename=work.temp  replace;
	data percentmet;
		set temp;
		length fv $20.;
		if variable=2 then fv='PercentVegRecMet'; else fv='PercentFruitRecMet';
		if _state ne 0 then statename=fipname(_state); else statename='National';
		keep _state statename fv nsum mean;
	run;

*********************************************************************************
 Estimate percent meeting recommendations				      
*********************************************************************************;

	*Use conversion equation regression coefficients to estimate predicted probability 
		of meeting the recommendation for each respondent;
	*Regression parameters for predicted probability of meeting fruit models;
	data parmsrecf;
		input sexagec model $26. Intercept f_juice_intake f_wholefrt_intake race1 race2 pir1 pir2;
		cards;
	1 Logit(Pr(yvar_f GE f_rec)) -6.06 1.8075 1.4916 0.3152 0.3618 -0.4813 -0.332
	2 Logit(Pr(yvar_f GE f_rec)) -4.6596 1.4798 1.6647 0.7545 0.2488 -0.5005 -0.3023
	3 Logit(Pr(yvar_f GE f_rec)) -4.0696 1.6626 1.5286 -0.105 -0.363 -0.5196 -0.6026
	4 Logit(Pr(yvar_f GE f_rec)) -5.4087 2.227 1.638 0.3825 -0.4413 0.0331 0.3026
	5 Logit(Pr(yvar_f GE f_rec)) -5.621 2.2377 1.5963 0.578 0.2255 -0.4155 -0.2661
	6 Logit(Pr(yvar_f GE f_rec)) -5.4948 2.1261 1.4769 0.5608 0.2916 0.0125 -0.054
	7 Logit(Pr(yvar_f GE f_rec)) -5.4824 2.0233 1.4799 0.0313 -0.1186 -0.3753 -0.6194
	8 Logit(Pr(yvar_f GE f_rec)) -5.2136 1.5826 1.5583 -0.324 -0.3004 -0.4122 -0.6227
;
	*Regression parameters for predicted probability of meeting fruit models;
	data parmsrecv;
		input sexagec model $26. Intercept v_leg_intake v_drkgr_intake v_orange_intake v_other_intake race1
		race2 pir1 pir2;
		cards;
	1 Logit(Pr(yvar_v GE v_rec)) -10.6741 3.0741 2.9105 1.8676 2.4543 1.0553 -0.5016 -0.9968 -0.3261
	2 Logit(Pr(yvar_v GE v_rec)) -7.2551 1.7454 1.5974 1.4572 1.7268 0.3476 -0.6758 -0.765 -0.6004
	3 Logit(Pr(yvar_v GE v_rec)) -5.0954 1.6233 1.4143 0.9277 1.5122 -0.2125 -0.69 -0.7254 -0.4627
	4 Logit(Pr(yvar_v GE v_rec)) -7.3053 1.8603 1.8111 1.2978 1.8868 -0.5384 -0.752 -0.7311 -0.4749
	5 Logit(Pr(yvar_v GE v_rec)) -8.8961 2.6907 0.8994 1.8565 2.3095 0.6188 -1.132 -0.4317 0.0101
	6 Logit(Pr(yvar_v GE v_rec)) -7.223 2.3002 1.7541 1.1451 1.9565 -0.059 -1.3282 -0.4398 -0.4741
	7 Logit(Pr(yvar_v GE v_rec)) -5.7497 1.9619 1.4864 0.8856 1.599 -0.026 -0.5869 -0.8123 -0.5782
	8 Logit(Pr(yvar_v GE v_rec)) -7.839 2.7171 2.3613 0.9653 1.7729 -0.8822 -1.3303 -0.566 -0.1485
;
run;
	proc sql;     
		create table three as
		select a._state, a.sexagec, a.FTJUDA1_, a.FRUTDA1_, a.BEANDAY_,
			a.GRENDAY_, a.ORNGDAY_, a.VEGEDA1_,_PSU, _STSTR, _LLCPWT, a.complete,
				b.intercept + (b.f_juice_intake*FTJUDA1_) 
				+ (b.f_wholefrt_intake*FRUTDA1_) + (b.race1*a.race1)
				+ (b.race2*a.race2)+ (b.pir1*a.pir1) + (b.pir2*a.pir2)
				as logitf label='log odds of predicted probability of meeting fruit recommendation',

			c.intercept + (c.v_leg_intake*BEANDAY_) 
				+ (c.v_drkgr_intake*GRENDAY_) +(c.v_orange_intake*ORNGDAY_) 
				+ (c.v_other_intake*VEGEDA1_)+ (c.race1*a.race1)  
				+ (c.race2*a.race2)+ (c.pir1*a.pir1) + (c.pir2*a.pir2)
				as logitv label='log odds of predicted probability of meeting veg recommendation',

			exp(calculated logitf)/(1+exp(calculated logitf)) 
				as pctf label='Predicted probability of meeting fruit recommendation',
	
			exp(calculated logitv)/(1+exp(calculated logitv)) 
				as pctv label='Predicted probability of meeting veg recommendation'
		from one as a,  parmsrecf as b, parmsrecv as c
		where a.sexagec=b.sexagec=c.sexagec
		order by a.sexagec;
	quit;

	*SAS Callable SUDAAN code to estimate national and state percent meeting recommendations;
	proc sort data=three; by _STSTR _PSU;
	proc descript data=three filetype=sas design=wr;
		NEST  _STSTR _PSU/missunit;
		weight _LLCPWT;
		subpopn complete=1/name="Respondents with no missing fv or demographic information";
		var pctf pctv ;
		tables 	_state ;
		class _state  ;
		output nsum mean /filetype=sas filename=work.temp  replace;
	data percents;
		set temp;
		length fv $20.;
		if variable=2 then fv='PercentVegetable'; else fv='PercentFruit';
		if _state ne 0 then statename=fipname(_state); else statename='National';
		keep _state statename fv nsum mean;
	run;
/********************************************************************************/
/* Merge output datasets with fruit and vegetable intake estimates to create	*/
/* one table with all 6 fv indicators overall and by state with sample sizes	*/
/********************************************************************************/

	data nsum (keep= _state statename nsum) estimates;
		set times_per_day percentmet(rename=(mean=median)) percents(rename=(mean=median));
	proc sort data=nsum; by _state;
	proc sort data=estimates; by _state;
	proc transpose data=estimates out=estimates;
		by _state;
		id fv;
		var median;
	data nsum;
		set nsum;
		by _state;
		if first._state;run;
	data estimates;
		merge nsum estimates;
		by _state;
		format timesfruit timesvegetable 8.1 PercentVegRecMet PercentFruitRecMet 
			PercentVegetable PercentFruit percent8.1 ;
		drop _name_ _label_;
	proc print data=estimates;
		title 'Fruit And Vegetable Intake, Behavioral Risk Factor Surveillance System, United States, 2011a';
	run;

